KARE MATRİSİN TERSİNİN BULUNMASI PROBLEMİNE PAN-REIFF YAKLAŞIMI

Yrd. Doç.Dr. Hasan DURUCASU
Anadolu Üniversitesi
İktisadi ve İdari Bilimler Fakültesi
İşletme Bölümü

ABSTRACT

Adjoint Matrix and Gauss-Jordan methods are the main approaches for finding the inverse of squared matrix. Adjoint Matrix method is not applicable for programming. However, since the Gauss-Jordan method comprises so many iterative procedures, it needs to much machine time. In this study, the revision of the fundamental matrix procedures which are related to basis matrix procedures, step by step, followed by the listing of developed related Basic subroutines. Thus, Pan-Reiff method which is an efficient approach to find inverse matrix has been introduced to the readers.

ÖZET

Kare matrisin tersinin bulunması için kullanılan başlıca yaklaşımlar Ek Matris ve Gauss-Jordan yöntemleridir. Bunlardan Ek Matris yönteminin programlamaya elverişli değildir. Buna karşılık Gauss-Jordan yöntemi ise çok sayıda tekrarlı işlem içerdiğinden, bilgisayara uygulandığında fazla işlem zamanına gereksinim duymaktadır. Bu çalışmada Pan-Reiff yönteminde kullanılan temel matris işlemlerinin adım adım ele alınıp gözden geçirilmesini ilgili Basic altyordamlarının listelenmesi izlemiştir. Böylelikle, ters matrisin bulunması problemine etkin bir yaklaşım olarak Pan-Reiff yöntemi okuyucuya tanıtılmıştır.

GİRİŞ

İşletme ve iktisat konusundaki olayları betimlemek için çok sayıda değişkene ihtiyaç duyulur. Değişken sayısının ve buna bağlı olarak veri miktarının artması durumunda kurulan matematik modeller genellikle doğrusal denklem sistemleriyle ifade edilmeye çalışılır. Cebirsel gösterimlerle çok karışık olan modeller matrisler yardımıyla kısa ve açık olarak ifade edilebilir. Öte yandan bilinen yaklaşımlarla yığın veri içeren modellerin çözümü güçlükler çıkarabilir. Bu tür problemlerin matrisler kullanılarak modellemesine gidilebildiği gibi çözümleri de matris işlemleri yardımıyla elde edilebilir.

Doğrusal denklem sistemleri çözüm uygulamalarında matris tersi alma işlemine sıklıkla başvurulmaktadır. Ters matrisin hesaplanmasına ilişkin klasik yöntemler, Ek Matris ve Gauss-Jordan Yöntemleridir.

Söz konusu bu yöntemlerden Ek Matris Yöntemi; elle uygulandığında boyutu üçten büyük kare matrislerde tekrarlı işlemlerin can sıkıcılığından; bilgisayarla denendiğinde programlanmasının basit olmadığından kullanımı elverişli değildir. Gauss-Jordan Yöntemi ise programlanmasının kolay olmasına karşın n^3 mertebesindeki çok sayıda hesaplamanın getirdiği işlem zamanı nedeni ile n büyük olduğunda tercih edilmemektedir.

Tanınmış bu iki yöntemin sıralanan sakıncaları göz önüne alınarak çalışmada, bu iki yönteme göre daha etkin olan Pan-Reiff Kare Matris Tersi Alma yöntemi ele alınmıştır.

MATRIS

Bilindiği gibi matris bazı toplama ve çarpım kurallarına göre satır ve sütun düzeninde gösterilen sayılar (ya da fonksiyonlar) kümesidir. Matris

$$\begin{bmatrix} a_{11} & a_{12} & \cdots & a_{1N} \\ a_{21} & a_{22} & \cdots & a_{2N} \\ \cdots & \cdots & \cdots & \cdots \\ a_{M1} & a_{M2} & \cdots & a_{MN} \end{bmatrix}$$ biçiminde bir tablo ile gösterilir.

Yukarıdaki biçiminde gösterilen tablonun yatay alt kümeleri satır, düşey alt kümeleri ise sütun adını alır. Yukarıdaki tablo simgesel olarak A ya da $[A]_{MxN}$ biçiminde ifade edilir. Son gösterimde yer alan MxN (satır sayısı x sütun sayısı) çifti matrisin boyutudur. Matrisin satır ve sütunlarının kesişim noktalarında yer alan a_{ij} 'ler ($i = 1,2,.....,M$; $j = 1,2,....,N$) matrisin elemanlarıdır.

KARE MATRİS

Satır sayısı sütun sayısına eşit olan ($M = N$) matrise kare matris adı verilir ve $[A]_{NxN}$ olarak gösterilir. NxN olan bir kare matrisin boyutu bir tek sayı ile (N) olarak belirlenir.
Kare matrisin boyutunu okumaya yönelik program kesimi aşağıdadır.

```
1003 REM BOYUT GIRISI ALTYORDAMI
1005 READ N
1006 RETURN
```

Ana programda yukarıda verilen altyordama veri sağlayan satır

```
9005 DATA 3
```

biçiminde düzenlenebilir.

MATRİSİN ELEMANLARINA DEĞER ATAMA

Kare matrisin a_{ij} elemanlarına değer atamaya yönelik altyordam kesimi aşağıda verilmiştir.

```
1010 KARE MATRISIN ELEMANLARININ DEGERLERININ SATIR
SATIR OKUTULMASI
1015 FOR I=1 TO N
1020 FOR J=1 TO N: READ A (I,J) : NEXT J
1025 NEXT I
1030 RETURN
```

Bu altyordama ilgili elemanların değerleri , ana programda yer alan

```
9010 DATA 1,1,1
9015 DATA 0,2,3
9020 DATA 5,5,1
```

satırlarıyla sağlanmaktadır.

MATRİSİN EKRANDA GÖRÜNTÜLENMESİ

Elemanlarının değerleri bilgisayarın belleğinde kayıtlı bulunan $[A]_{NxN}$ kare matrisinin ekranda alışılmış matris biçiminde görüntülenmesini aşağıda verilen altyordam sağlar.

```
1050 REM A MATRISINI GORUNTULEME ALTYORDAMI
1055 FOR I=1 TO N
1060 FOR J=1 TO N: PRINT USING "##.###"; A (I,J);: NEXT J
1065 PRINT
1070 NEXT I
1080 RETURN
```

SIFIR MATRİS

Bir matrisin bütün elemanlarının değerleri sıfır ise bu matrise Sıfır Matris adı verilir.

MATRİS TOPLAMI

İki matrisin toplanabilir olmasının ön koşulu bu iki matrisin aynı boyutta olmasıdır. Örneğin aynı boyutlara sahip *BIRIM* ve *E* kare matrisleri toplanabilir. Sonuç matrisi *D* de, işleme konu olan *BIRIM* ve *E* matrisleriyle aynı boyuttadır.

Matris toplamı işlemi matris yazılımı kullanılarak

$$[D]_{NxN} = [BIRIM]_{NxN} + [E]_{NxN}$$

biçiminde yazılırken, matrisin elemanları cinsinden

$$d_{ij} = birim_{ij} + e_{ij} \qquad (i = 1,\dots,N \qquad j = 1,\dots,N)$$

olarak ifade edilir. Matris toplama işlemine ilişkin altyordam aşağıda verilmiştir.

```
1150 REM MATRIS TOPLAMA ALTYORDAMI :D=BIRIM+E
1155 FOR I=1 TO N
1160 FOR J=1 TO N: D (I,J) =BIRIM (I,J) +E (I,J) : NEXT J
1170 NEXT I
1175 RETURN
```

MATRİS FARKI

Aynı boyutlara sahip matrislerin farkı da alınabilir. Sonuç matrisi E, fark işlemine konu olan $A1$ ve $B1$ matrisleriyle aynı boyuttadır. Matris farkı işlemleri matris yazılımı kullanılarak $[E]_{NxN} = [A1]_{NxN} - [B1]_{NxN}$ biçiminde yazılırken, matrisin elemanları cinsinden bu ifade

$$e_{ij} = a1_{ij} - b1_{ij} \quad (i = 1,...,N \qquad j = 1,...,N)$$

biçimindedir. Aşağıda verilen altyordam matris farkı almaya yöneliktir.

```
1200 REM MATRIS FARKI ALMA ALTYORDAMI :E=A1 -B1
1205 FOR I=1 TO N
1210 FOR J=1 TO N: E (I,J) =A1 (I,J) -B1 (I,J) : NEXT J
1220 NEXT I
1230 RETURN
```

MATRİS ÇARPIMI

C çarpım matrisinin c_{ij} elemanı, $A1$ matrisinin i. satırının elemanlarıyla $B1$ matrisinin j. sütun elemanlarının çarpımlarının toplamı olduğundan $A1$ x $B1$ matris çarpımın oluşturulabilmesi için, $A1$ matrisinin sütun sayısının $B1$ matrisinin satır sayısına eşit olması gerekir.

Matris çarpım işlemi matris yazılımı kullanılarak $[A1]_{LxN} x[B1]_{NxM} = [C]_{LxM}$ biçiminde yazılır. C çarpım matrisinin her bir c_{ij} elemanı ise $c_{ij} = \sum_{k=1}^{N} a1_{ik} * b1_{kj}$ olarak bulunur. İlgili altyordam aşağıda yeralmaktadır.

```
1250 REM MATRIS CARPMA ALTYORDAMI :C=A1*B1
1255 FOR I=1 TO N
1260 FOR J=1 TO N: C (I,J) =0
1265 FOR K=1 TO N
1270 C (I,J) =C (I,J) +A1 (I,K) *B1 (K,J)
1275 NEXT K
1280 NEXT J
1285 NEXT I
1290 RETURN
```

BİRİM MATRİS

Esas köşegen elemanlarının değeri 1 , diğer elemanlarının değeri 0 olan kare matrisin birim matris olduğu bilinmektedir. Birim matrisin uygun boyuttaki bir A matrisiyle (soldan ya da sağdan) çarpımı yine A matrisini verir. Diğer bir anlatımla birim matris, matris çarpma işleminin etkisiz elemanıdır.

N boyutundaki bir $BIRIM$ matrisini bellekte oluşturmaya yönelik altyordam aşağıda verilmiştir.

```
1400 REM BIRIM MATRIS OLUSTURMA
1410 FOR I=1 TO N
1415 FOR J=1 TO N:BIRIM (I,J) =0
1420 IF I=J THEN BIRIM (I,J) =1
1425 NEXT J
1430 NEXT I
1435 RETURN
```

TERS MATRİS

$BIRIM$ birim matris olmak üzere $A^{-1}xA = BIRIM$ bağıntısını sağlayan A^{-1} matrisine A matrisinin tersi denir.

NORM

Bilindiği gibi matrisin bir tek sayı ile nicelleştirilmesini norm adı verilen büyüklük sağlar.

Satır normu, matrisin satırlarının elemanlarının mutlak değerleri toplamının en büyüğüdür. Sıfır matrisin satır normunun da 0 olacağı açıktır.

A1 kare matrisine ilişkin satır normu hesaplamaya yönelik altyordam aşağıda verilmiştir.

```
1300 REM SATIR NORMU HESAPLAMA ALTYORDAMI
1305 ENBUYUK=0
1310 FOR I=1 TO N
1315 ENBUYUK2=0
1320 FOR J=1 TO N
1325 ENBUYUK2=ENBUYUK2+ABS (A1 (I,J) )
1330 NEXT J
1335 IF ENBUYUK2>ENBUYUK THEN ENBUYUK=ENBUYUK2
```

```
1340 NEXT I
1345 SATNORM=ENBUYUK
1347 RETURN
```

Benzer biçimde sütün normu ise matrisin sütun elemanlarının mutlak değerleri toplamının en büyüğüdür. A kare matrisine ilişkin sütun normu değerini hesaplayan altyordam izleyen kesimde yer almaktadır.

```
1350 REM SUTUN NORMU HESAPLAMA ALTYORDAMI
1355 ENBUYUK=0
1360 FOR I=1 TO N
1365 ENBUYUK2=0
1370 FOR J=1 TO N
1375 ENBUYUK2=ENBUYUK2+ABS (A(J,I) )
1380 NEXT J
1385 IF ENBUYUK2>ENBUYUK THEN ENBUYUK=ENBUYUK2
1390 NEXT I
1395 SUTNORM=ENBUYUK
1399 RETURN
```

EVRİK MATRİS

A matrisinin evriği olarak A^T olarak gösterilen ve A matrisinin satır elemanlarıyla sütun elemanlarının yer değiştirmesiyle oluşturulan matris kullanılır. Böylelikle A^T matrisinin her bir elemanı

$$a_{ij}^T \,,\; a_{ij}^T = a_{ji} \qquad (i = 1,\dots,N \;\; j = 1,\dots,N\,)$$

olarak yazılır.

Programda A^T matrisi, ilk B matrisinin oluşturulması altyordamında B matrisi olarak ifade edilmiştir.

BAŞLANGIÇ TERS MATRİSİ

Yinelenen işlemlerden oluşan Pan-Reiff yönteminde, sürecin yakınsaklığını garantilemek amacıyla A matrisinin tersi olarak B matrisi için geçerli ilk yaklaşık matrisin ortaya konması gereklidir. B matrisine ilişkin ilk yaklaşık matris elde etme, Pan-Reiff tarafından A matrisinin evriğini alma ve her bir elemanı satır normu ve sütun normu çarpımına bölme olarak belirlenmiştir. Böylesi bir B matrisinin oluşturulmasına yönelik altyordam aşağıda verilmiştir.

```
1500 REM ILK B MATRISININ OLUSTURULMASI
1501 FOR I=1 TO N
1502 FOR J=1 TO N: A1(I,J)=A(I,J) : NEXT J
1503 NEXT I
1504 GOSUB 1300:
REM SATIR NORMU HESAPLAMA ALTYORDAMI
1505 GOSUB 1350:
REM SUTUN NORMU HESAPLAMA ALTYORDAMI
1509 T=SATNORM*SUTNORM
1510 IF T=0 THEN PRINT "SIFIR MATRIS": GOTO 1855
1515 FOR I=1 TO N
1520 FOR J=1 TO N: B(J,I) =A (I,J) /T: NEXT J
1525 NEXT I
1530 RETURN
```

Büyük bir olasılıkla A matrisinin tersi olan A^{-1} matrisine ilk yaklaşım olarak yukarıda ortaya konan ilk B matrisi A^{-1} matrisine eşit almayacak ancak A^{-1}'e bir yaklaşım olacaktır. Bu nedenle elde edilmesi arzulanan A^{-1}'e daha yakın yeni bir yaklaşık değerin tekrar eden işlem adımları yardımıyla hesaplanması yoluna gidilecektir.

A matrisinin tersi olan A^{-1} matrisine ilk yaklaşım olarak ortaya konan B matrisi A^{-1} matrisine eşit olmayacağından $BxA = BIRIM$ eşitliği sağlanamayacaktır. Elde edilmesi istenen $A^{-1}xA$ ile halen elde olan BxA arasındaki fark,

$$A^{-1}xA - BxA = E$$

olarak ifade edilebilir.

HATA MATRİSİ

İlk B matrisinin yukarıda belirtilen şekilde hesaplanmasının ardından, yukarıdaki son eşitlikten, $BIRIM$ birim matrisi ile $B \times A$ matris çarpımının farkı alınıp E hata matrisi $E = BIRIM - BxA$ olarak oluşturulur. E hata matrisinin hesaplanmasına ilişkin altyordam izleyen kesimde yer almaktadır.

```
1550 REM E MATRISININ HESAPLANMASI
1555 FOR I=1 TO N
1560 FOR J=1 TO N: A1 (I,J) =B (I,J) :B1 (I,J) =A (I,J) : NEXT J
```

```
1565 NEXT I
1570 GOSUB 1250: REM MATRIS CARPMA ALTYORDAMI
1575 FOR I=1 TO N
1577 FOR J=1 TO N: A1 (I,J) =BIRIM (I,J) :B1 (I,J) =C (I,J) : NEXT J
1579 NEXT I
1580 GOSUB 1200: REM MATRIS FARKI ALMA ALTYORDAMI
1585 RETURN
```

Pan-Reiff yönteminin temeli, E hata matrisinin sıfır matrise eşit kılınıncaya ya da sıfır matrise kabul edilebilir yakınlıkta bir düzeye indirgeninceye değin, yeni B matrislerinin hesaplanması işlem adımlarının tekrar edilmesine dayanır.

YENİ B MATRİSİNİN HESAPLANMASI

Pan-Reiff yönteminde ilk B değeri ortaya konulup, buna karşılık gelen E hata matrisi hesaplandıktan sonra, B'nin yeni değeri $B = Bx(BIRIM + E)$ eşitliği yardımıyla hesaplanmaktadır. Bu işlemi yerine getiren altyordam aşağıda verilmiştir.

```
1600 REM DONGU ICINDE YENI B LERIN HESAPLANMASI
1604 GOSUB 1150: REM MATRIS TOPLAMA ALTYORDAMI
1605 FOR I=1 TO N
1610 FOR J=1 TO N: A1 (I,J) =D (I,J) :B1 (I,J) =B (I,J) : NEXT J
1615 NEXT I
1620 GOSUB 1250: REM MATRIS CARPMA ALTYORDAMI
1625 FOR I=1 TO N
1630 FOR J=1 TO N: B (I,J) =C (I,J) : NEXT J
1635 NEXT I
1640 RETURN
```

B TERS MATRİSİNİN HESAPLANMASINA İLİŞKİN ARDIŞIK İŞLEM ADIMLARI

A kare matrisinin tersi A^{-1} matrisini bulmaya yönelik yukarıda sıralanan işlem adımları

1. Başlangıç ters matrisi B'yi oluşturma,
2. E hata matrisini hesaplama,
3. Yeni ters matris B'yi hesaplama,

4. E'nin satır normunun yeterince küçük(örneğin 10^{-6}) kılınmasına değin 2. ve 3. adımların tekrar uygulanması

biçiminde özetlenebilir.

İlgili yayınlarda 4.'teki ölçütün 5. işlem adımından önce dikkate alınmaması önerilmektedir.

Ayrı ayrı açıklanan ve yukarıda özetlenen işlem adımlarını esas alıp, verilen bir kare matrise ilişkin ters matris bulma altyordamı aşağıda verilmiştir.

```
1650 REM TERS MATRIS HESAPLAMA ALTYORDAMI
1651 GOSUB 1400: REM BIRIM MATRIS OLUSTURMA
1655 GOSUB 1500: REM İLK B MATRISININ OLUSTURULMASI ALTYORDAMI
1660 GOSUB 1550: REM E MATRISININ HESAPLANMASI
1665 FOR I=1 TO N
1666 FOR J=1 TO N: A1 (I,J) =E( I,J) : NEXT J
1668 NEXT I
1670 GOSUB 1300: REM SATIR NORMU HESAPLAMA ALT YORDAMI
1675 ESKI=SATNORM
1677 YENI=ESKI
1679 SAYAC=0
1690 ESKI=YENI
1695 GOSUB 1600: REM DONGU ICINDE YENI B LERIN HESAPLANMASI ALTYORDAMI
1700 GOSUB 1550: REM E MATRISININ HESAPLANMASI
1705 FOR I=1 TO N
1710 FOR J=1 TO N: A1 (I,J) =E (I,J) : NEXT J
1715 NEXT I
1720 GOSUB 1300: REM SATIR NORMU HESAPLAMA ALTYORDAMI
1725 YENI=SATNORM
1730 SAYAC=SAYAC+1
1735 IF YENI> .000001 OR YENI<ESKI AND SAYAC<5 THEN GOTO 1690
1745 IF YENI> .000001 THEN PRINT "TERS YOK          ": STOP
1755 RETURN
```

TERS MATRİS BULMA ANA PROGRAMI

Pan-Reiff yöntemini esas alıp, tanıtılan altyordamları kullanarak verilen 3x3'lük bir A kare matrisinin tersi A^{-1} matrisini bulmaya yönelik ana program ve örnek çıktı izleyen kesimde verilmiştir.

```
100 REM PAN-REIFF ALGORITMASI UYARINCA TERS MATRIS BULAN BASIC PROGRAMI
103 GOSUB 1003: REM KARE MATRISIN BOYUTUNUN OKUNMASI
105 GOSUB 1010: REM KARE MATRISIN ELEMANLARININ DEGERLERININ SATIR SATIR OKUTULMASI
107 PRINT "A MATRISI            :"
109 PRINT "---------------------------------"
110 GOSUB 1050: REM MATRIS GORUNTULEME ALTYORDAMI
120 GOSUB 1650: REM TERS MATRIS HESAPLAMA ALTYORDAMI
125 PRINT "A MATRISININ TERSI:"
130 PRINT "-----------------------------"
135 FOR I=1 TO N
140 FOR J=1 TO N :A (I,J)=B (I,J) :NEXT J
145 NEXT I
150 GOSUB 1050: REM MATRIS GORUNTULEME ALTYORDAMI
205 STOP
...
1855 END
```

```
A MATRISI
---------------------------------
 1.000      1.000      1.000
 0.000      2.000      3.000
 5.000      5.000      1.000
A MATRISININ TERSI       :
---------------------------------
 1.625     -0.500     -0.125
-1.875      0.500      0.375
 1.250     -0.000     -0.250
```

Basic programları artan satır numarası esasına göre bilgisayar belleğine yerleşirler. Bu nedenle bu çalışmada yer alan program denenmek istendiğinde, tüm kesimlerde yer alan program satırları bilgisayara aktarılmalıdır.

Farklı A kare matrislerinin terslerinin bulunabilmesi için DATA satırlarındaki değerler değiştirilmelidir.

Yukarıda verilen örnek programda, indisli değişkenler için yazılımın otomatik boyutlandırma varsayım değerleriyle yetinilmiştir. 10'dan büyük boyutlu A kare matrisleri için işlem yapılmak istendiğinde, programda yer alan tüm indisli değişkenler DIMENSION deyimi kullanılarak uygun biçimde boyutlandırılmalıdır.

YARARLANILAN KAYNAKLAR

AYTAÇ M., SEVÜKTEKİN M., IŞIĞIÇOK E, **Sosyal Bilimlerde Matematik**, (Ezgi Kitapevi: Bursa), 1998.

ÇAKMAK Z., **İktisadi ve İdari Bilimlerde Matematik II**, (Ekspres Matbaası: Kütahya), 1998.

ERDOĞAN N.K., **İşletme ve Ekonomi Uygulamalı Matematik II**, (Birlik Ofset: Eskişehir), 1998.

FADDEEV D. K., (Çev: WILLIAMS R. C.), **Computational Methods of Linear Algebra**, (W. H. Freeman: San Fransisco), 1963.

FINKBEINER D. T., **Introduction to Matrices and Linear Transformation**, (W. H. Freeman: San Francisco), 1966.

HACISALİHOĞLU H. H., **Lineer Cebir**, (Gazi Üniversitesi Fen-Edebiyat Fakultesi Yayınları: Ankara), 1980.

HÜSEYİN Ö., SEZER E., **Elementary Matrix Algebra and Differential Equations**, (Metu: Ankara), 1980.

www.chez.com/algor/pan.htm